# Maintainer: Filipe Laíns (FFY00) <lains@archlinux.org>
# Contributor: Michal Krenek (Mikos) <m.krenek@gmail.com>

pkgname=python-qt.py
_pkgname=Qt.py
pkgver=1.3.8
pkgrel=2
pkgdesc='Python compatibility wrapper around all Qt bindings'
arch=('any')
url='https://github.com/mottosso/Qt.py'
license=('MIT')
depends=('python' 'qt5-python-bindings')
makedepends=('python-setuptools')
source=("$pkgname-$pkgver.tar.gz::$url/archive/$pkgver.tar.gz")
sha512sums=('3d584e1574bbb9458847e44ab4c901cc134d1669c549cfff3c8ac391f3dca87f48166e874305febb5092dea0d8baa58bd66fae71c100aa9b29b089502d341f71')

build() {
  cd $_pkgname-$pkgver

  python setup.py build
}

package() {
  cd $_pkgname-$pkgver

  python setup.py install --root="$pkgdir/" --optimize=1 --skip-build

  install -Dm 644 LICENSE "$pkgdir"/usr/share/licenses/$pkgname/LICENSE
}
